"Lunch on the nature"

A game
1 The rules

There are three cannibals and three preachers on the right side of a river. They need to cross the river using one boat. The boat can carry not more than two persons. The principal restriction is the following. Since the cannibals are very angry and very hungry they permanently think about a way to eat the preachers. To avoid that, the number of preachers on each side (if not zero) always has to be more than the number of cannibals.

2 The model
We formalize the rules in the following way.

2.1 The state

The system state is characterized by the four integer variables: CannibalsOnTheLeft, CannibalsOnTheRight, PreachersOnTheLeft and PreachersOnTheRight with the obvious meaning. Initially all the participants are on the right bank of the river.
var CannibalsOnTheLeft  as Integer = 0

var PreachersOnTheLeft  as Integer = 0

var CannibalsOnTheRight as Integer = 3

var PreachersOnTheRight as Integer = 3

Also, we need a variable to indicate where the boat is. 
enum RiverSide

  Left
  Right
Initially it is on the right.
var BoatPosition as RiverSide = Right 

2.2 The transitions
The two main game steps are MoveRight and MoveLeft with the natural preconditions. 

procedure MoveRight( c as Integer, p as Integer )

  require ( BoatPosition = Left ) 

    and ( CannibalsOnTheLeft >= c )

    and ( PreachersOnTheLeft >= p )

    and ( c+p in {1,2} )
    and not ReadyForLunch()
  CannibalsOnTheLeft  -= c
  CannibalsOnTheRight += c
  PreachersOnTheLeft  -= p

  PreachersOnTheRight += p
  BoatPosition := Right
The boat has to be on the corresponding side, the number of the travelers should be appropriate, and the final condition RadyForLunch() should be false.
procedure MoveLeft( c as Integer, p as Integer )

  require ( BoatPosition = Right ) 

    and ( CannibalsOnTheRight >= c )

    and ( PreachersOnTheRight >= p )

    and ( c+p in {1,2} )

    and not ReadyForLunch()

  CannibalsOnTheLeft  += c
  CannibalsOnTheRight -= c
  PreachersOnTheLeft  += p

  PreachersOnTheRight -= p
  BoatPosition := Left
If due to a wrong step the principal restriction is violated, the cannibals accomplish the wild lunch. 
procedure Lunch()

  require WildLunch()

  if (CannibalsOnTheLeft > PreachersOnTheLeft) and (PreachersOnTheLeft > 0)
    PreachersOnTheLeft := 0
  elseif (CannibalsOnTheRight > PreachersOnTheRight) 

                                   and (PreachersOnTheRight > 0) 
    PreachersOnTheRight := 0
On the other hand, the victory condition is formalized in the HappyLunch().

function ReadyForLunch() as Boolean

  return WildLunch() or HappyLunch()
function WildLunch() as Boolean

  return (CannibalsOnTheLeft > PreachersOnTheLeft) and (PreachersOnTheLeft > 0) 

   or (CannibalsOnTheRight > PreachersOnTheRight) and (PreachersOnTheRight > 0)

   or (PreachersOnTheRight+PreachersOnTheLeft < 3)

function HappyLunch() as Boolean
  return (PreachersOnTheLeft = 3) and (CannibalsOnTheLeft = 3)
2.3 Simulation
We can execute the model with a random scenario.

Main()

  step until ReadyForLunch()
    try     
      RandomAction()

      ShowState()

    catch e: skip 

  step
    RandomAction()    
    ShowState()
  step
    ShowState()
    WriteLine(if HappyLunch() then "Victory" else "Lunch") 
procedure RandomAction()

  if ReadyForLunch() then
    Lunch()
  else
    choose p in {0,1,2},
           c in {0,1,2} where p+c in {1,2}
      if BoatPosition = Left then
        MoveRight(c,p)

      else
        MoveLeft(c,p)

procedure ShowState()

  WriteLine("Left:"  + ToString( [ CannibalsOnTheLeft, PreachersOnTheLeft ] ) +

          " Right:" + ToString( [ CannibalsOnTheRight, PreachersOnTheRight ] )+

          " Boat:" + ToString( BoatPosition ) )
2.4 Analysis 

Using the AsmL Tester Tool we can explore completely the state space (for this game it is rather small). The numbers on the picture stand for the state variables. The initial state is located on the left (see the START label).

